
Лабораторная работа №2. Вычисление по формуле 

2.1. Цели работы 
1. Ознакомиться с арифметическими инструкциями ассемблера. 

2.2. Используемое программное обеспечение 
1. GNU Binutils. 
2. Клиент сетевой системы для проведения лабораторных работ asm2.jar. 

2.3. Порядок выполнения лабораторной работы 
1. Зарегистрируйтесь в системе под своим именем. 
2. Ознакомьтесь с заданием по лабораторной работе. 
3. Напишите программу, соответствующую вашему заданию. 
4. Нажмите кнопку Проверить и выберите файл с исходным кодом. 

2.4. Задание 
После успешной регистрации в системе для проведения лабораторных работ, Вы получите 
индивидуальное задание. Пример задания приведён на рис. 2.1. 

 

Рис. 2.1. Пример задания 

2.5. Пример выполнения задания 
Рассмотрим пример задания, приведённый в пункте 2.4. Требуется прочитать данные со 
стандартного ввода, вычислить значение выражения для данного ввода и записать младший 
десятичный разряд полученного результата на стандартный вывод. Входные данные передаются в 
формате ABCD без каких либо разделительных символов. Каждое число лежит в диапазоне 0..9. 
Для приведённых входных данных 4766 переменные имеют следующие значения: 

A = 4 
B = 7 
C = 6 
D = 6 
Подставив эти значения в выражение ((C * (D + C)) * (A * D)), получим результат: 

((6 * (6 + 6)) * (4 * 6)) = 1728 

В младшем десятичном разряде получаем цифру 8. 

Представим выражение в качестве последовательности арифметических действий: 



1. Сложить D и C. 
2. Результат умножить на C. 
3. Результат умножить на A. 
4. Результат умножить на D. 

Пример программы на ассемблере для 32-битного окружения Mac OS X с подробными 
комментариями приведён в листинге 2.1. 

Листинг 2.1 

.intel_syntax noprefix 
 
SYS_EXIT = 1 
SYS_READ = 3 
SYS_WRITE = 4 
 
STDIN = 0 
STDOUT = 1 
STDERR = 2 
 
/* Секция данных */ 
.data 
 
/* Сообщение об ошибке */ 
error_msg: 
.ascii "Error reading STDIN\n" 
 
/* Симаол конца строки */ 
end_of_line: 
.ascii "\n" 
 
/* Выделяем для входных данных 4 байта */ 
buf: 
.space 4, 0 
 
/* Секция кода */ 
.text 
 
.globl start 
 
/* Входная точка программы */ 
start: 
 
/* Прочитаем со стандартного ввода значения переменных */ 
push 4 
push offset buf 
push STDIN 
mov eax, SYS_READ 
push eax 
int 0x80 
add esp, 16 
 
/* В случае ошибки чтения выводим сообщение и завершаем работу */ 
cmp eax, -1 
jz error 
 



/* Все переменные имеют разрядность одного символа (1 байт). Обнуляем 
регистр eax, записываем в младший байт регистра символьное 
представление значения переменной и вычитаем код символа 0, чтобы 
получить число. Регистр eax содержит значение переменной D */ 
xor eax, eax 
mov al, [buf + 3] 
sub al, '0 
 
/* Копируем значение переменной D в регистр ebx */ 
mov ebx, eax 
 
/* Обнуляем регистр edx и записываем в него значение переменной C */ 
xor edx, edx 
mov dl, [buf + 2] 
sub dl, '0 
 
/* Складываем переменные C и D. Результат сохраняется в регистре eax 
*/ 
add eax, edx 
 
/* Умножаем результат на переменную C */ 
imul edx 
 
/* Обнуляем регистр ecx и записываем в него значение переменной A */ 
xor ecx, ecx 
mov cl, [buf] 
sub cl, '0 
 
/* Умножаем результат предыдущего действия на переменную A */ 
imul ecx 
 
/* Умножаем результат на переменную D, сохранённую в регистре ebx */ 
imul ebx 
 
/* Делим результат выражения на 10, чтобы получить в остатке значение 
младшего десятичного разряда полученного результата */ 
mov ebx, 10 
xor edx, edx 
div ebx 
 
/* При делении остаток записывается в регистр edx. Прибавляем к нему 
код символа 0, чтобы получить символьное представление числа */ 
add dl, '0 
mov [buf], dl 
 
/* Записываем этот символ в стандартный вывод */ 
push 1 
push offset buf 
push STDOUT 
mov eax, SYS_WRITE 
push eax 
int 0x80 
add esp, 16 
 
/* Записываем в стандартный вывод символ перевода строки */ 
push 1 
push offset end_of_line 
push STDOUT 



mov eax, SYS_WRITE 
push eax 
int 0x80 
add esp, 16 
jmp exit 
 
error: 
push 20 
push offset error_msg 
push STDERR 
mov eax, SYS_WRITE 
push eax 
int 0x80 
add esp, 16 
 
/* Завершение программы */ 
exit: 
push 0 
mov eax, SYS_EXIT 
push eax 
int 0x80 
add esp, 8 

